
1 [bookmark: _Toc122434485][bookmark: _Toc126845572]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of NR5GC test case 8.2.6.2.2  which is part of the NR5GC test suite in iwd-TTCN3-B2022-09_D22wk49.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log file(s) is (are) provided as evidence. 
Contact:	Parikshit Bhise
	parikshit.bhise@rohde-schwarz.com 
	
1. [bookmark: _Toc126845573]Verification Test Summary 
Test Case: 	8.2.6.2.2
ATS Version:	iwd-TTCN3-B2022-09_D22wk49
[bookmark: _Hlk37919671]System Simulator used:	R&S® 5G Protocol Conformance Test platform
UE used:	Snapdragon X70 5G Modem-RF System, Qualcomm Technologies’ 5th generation modem-to-antenna 5G solution
Verification Status:	PASS

[bookmark: _Toc122434488][bookmark: _Toc295288959]

2 [bookmark: _Toc126845574]Corrections required
2.1 [bookmark: _Toc126845575]Function f_NR5GC_RRC_ProcedureDelay ()
	Function name
	f_TC_8_2_6_2_2_NR5GC_TestBody ()

	Reason for change
	1) In the current TTCN Implementation , handling of Pdu session modification complete is done outside of the function f_NR5GC_RRC_ProcedureDelay()

This behaviour is not working stably with different UE behaviours . 

(a) Some UEs are too quick in sending the Pdu Session Modification Complete along with reconfiguration complete 

(b) As the grant is periodic inside the function f_NR5GC_RRC_ProcedureDelay
, before stopping the grant , there could be corner case of sending second UL grant which will help UE to send the Pdu Session Modification Complete , and causes the test case to fail ( unexpected Message received ) 

To avoid this race condition happening, It is proposed to handle the Pdu Session Modification Complete inside the function f_NR5GC_RRC_ProcedureDelay() in case of NR-DC ; 



	Summary of change
	1) To avoid the instability during the reception of Pdu Session Modification complete , it is proposed to handle the Pdu Session Modification Complete inside the function f_NR5GC_RRC_ProcedureDelay() by using the boolean flag nrdc_pdusession which is set only in case of 8.2.6.2.2 ( NR DC Procedure delay test case ) 


	TTCN module
	Common\NR5GC\NR5GC_RRCSteps.ttcn

	MCC160 Comment
	



Before Change:
	   function f_NR5GC_RRC_ProcedureDelay(NR_CellId_Type                        p_NR_CellId,
                                      template (value) NR_SRB_COMMON_REQ    p_SRB_COMMON_REQ,
                                      template (present) NR_SRB_COMMON_IND  p_SRB_COMMON_IND,
                                      integer                               p_Nms,
                                      charstring                            p_TestcaseStep,
                                      template (value) NR_FreqDomainSchedulExplicit_Type p_FreqDomainSchedul := cs_NR_FreqDomainSchedulExplicitUL_Def
                                      ) runs on NR_BASE_PTC return NR_SRB_COMMON_IND
  {
    var integer v_T1Slots;
    var integer v_T2Slots;
    var integer v_T3Slots;
    var integer v_T4Slots;
    var SubFrameTiming_Type v_T1; // Send DL Msg
    var SubFrameTiming_Type v_T2; // HARQ ACK receive
    var SubFrameTiming_Type v_T3; // Start UL Grant start
    var SubFrameTiming_Type v_T4; // UL RRC PDU received
    var SubFrameTiming_Type v_Tnow;
    var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
    var integer v_Delta1;
    var integer v_Delta2;
    var integer v_GrantPeriodicity;
    var integer v_DL_RetransmissionCnt := 0;
    var NR_SYSTEM_IND v_SYS_IND;
    var boolean v_T2Flag := false;
    var boolean v_T4Flag := false;
    var integer v_K1;
    var integer v_K2 := 4;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;     // Maximum number per SFN complete cycle
    var template (value) NR_SRB_COMMON_REQ v_SRB_COMMON_REQ := p_SRB_COMMON_REQ;

    timer t_Watchdog;
    
    //@sic R5s210842 sic@
    if (f_TimingInfo_IsNow(v_SRB_COMMON_REQ.Common.TimingInfo)) {
        v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
        v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;
    } else {
        v_T1 := valueof(v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame);
    }
    v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
    // Initialise GrantPeriodicity, Delta1, K1
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_GrantPeriodicity := 1; // Grant repeated in every slot
        v_Delta1 := 0;
        v_K1 := 7;
    } else { // TDD
        select(v_SCS) {
            case(kHz15) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
                v_K1 := 7;
            }
            case(kHz30) {
                v_GrantPeriodicity := 10; //Grant repeated every 10 slots
                v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
                v_K1 := 5;
            }
            case(kHz120) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
                v_K1 := 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    
    //Calculate T3
    v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;  //@sic R5s210842 change 2 sic@
    v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
    
    //T1: send DL Msg
    SRB.send(v_SRB_COMMON_REQ);

    //Enable HARQ ack/nack reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
    
    //T3 configure SS with periodic UL Grant at T3
    f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, p_FreqDomainSchedul, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
                                          
    // Start WatchDog timer
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
    t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 50) / 1000.0);  /* 50ms added to be at the safe side */
    
    alt {
        /* HARQ NACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
            v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
            if (v_DL_RetransmissionCnt > 4) {
                t_Watchdog.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
            } else {
                repeat;
            }
        }
        /* HARQ ACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
            v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
            f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable), -, tsc_NoCnfReq);    /* disable indication of HARQ ack/nack; @sic R5s220683 sic@ */
            v_T2Flag := true;
            repeat;
        }
        /* HARQ error indication */
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        [] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {
            v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
            v_T4Flag := true;
            t_Watchdog.stop;
        }
        [] t_Watchdog.timeout {                                                                                     /* timeout => after T5 */
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
        }
    }
    //Stop UL Grant
    f_NR_ULGrantConfiguration_Stop(p_NR_CellId);
    
    if (not v_T2Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
    }
    if (not v_T4Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL Msg not received");
    }
    
    //Check the Delay
    v_T2Slots := (((v_T2.SFN.Number * 10) + v_T2.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T2) + v_MaxNumberSlots;
    v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
    //Initialise Delta2
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_Delta2 := 0;
    } else { //TDD
        select(v_SCS) {
            case(kHz15) {
                v_Delta2 := (5 - (v_T2Slots - v_K1 + v_Nslots)) mod 5;
            }
            case(kHz30) {
                v_Delta2 := (4 - (v_T2Slots - v_K1 + v_Nslots) mod 10) mod 10;
            }
            case(kHz120) {
                v_Delta2 := (5- (v_T2Slots - v_K1 + v_Nslots) mod 5) mod 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    if ( ((v_T4Slots - v_K2) - (v_T2Slots - v_K1))mod v_MaxNumberSlots <= (v_Nslots + v_Delta2)) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
    }

   return v_ReceivedNR_SRB_COMMON_IND;
  }



After Change:
	import from NG_NAS_Common all; //WA#WI=988940
import from NG_NASTemplateFunctions all; //WA#WI=988940

function f_NR5GC_RRC_ProcedureDelay(NR_CellId_Type                        p_NR_CellId,
                                      template (value) NR_SRB_COMMON_REQ    p_SRB_COMMON_REQ,
                                      template (present) NR_SRB_COMMON_IND  p_SRB_COMMON_IND,
                                      integer                               p_Nms,
                                      charstring                            p_TestcaseStep,
                                      template (value) NR_FreqDomainSchedulExplicit_Type p_FreqDomainSchedul := cs_NR_FreqDomainSchedulExplicitUL_Def,
                                      template (omit) boolean nrdc_pdusession := omit //WA#WI=988940
                                      ) runs on NR_BASE_PTC return NR_SRB_COMMON_IND
  {
    var integer v_T1Slots;
    var integer v_T2Slots;
    var integer v_T3Slots;
    var integer v_T4Slots;
    var SubFrameTiming_Type v_T1; // Send DL Msg
    var SubFrameTiming_Type v_T2; // HARQ ACK receive
    var SubFrameTiming_Type v_T3; // Start UL Grant start
    var SubFrameTiming_Type v_T4; // UL RRC PDU received
    var SubFrameTiming_Type v_Tnow;
    var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
    var integer v_Delta1;
    var integer v_Delta2;
    var integer v_GrantPeriodicity;
    var integer v_DL_RetransmissionCnt := 0;
    var NR_SYSTEM_IND v_SYS_IND;
    var boolean v_T2Flag := false;
    var boolean v_T4Flag := false;
    var integer v_K1;
    var integer v_K2 := 4;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;     // Maximum number per SFN complete cycle
    var template (value) NR_SRB_COMMON_REQ v_SRB_COMMON_REQ := p_SRB_COMMON_REQ;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN); //WA#WI=988940

    timer t_Watchdog;
    
    //@sic R5s210842 sic@
    if (f_TimingInfo_IsNow(v_SRB_COMMON_REQ.Common.TimingInfo)) {
        v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
        v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;
    } else {
        v_T1 := valueof(v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame);
    }
    v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
    // Initialise GrantPeriodicity, Delta1, K1
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_GrantPeriodicity := 1; // Grant repeated in every slot
        v_Delta1 := 0;
        v_K1 := 7;
        } else { // TDD
        select(v_SCS) {
            case(kHz15) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
                v_K1 := 7;
            }
            case(kHz30) {
                v_GrantPeriodicity := 10; //Grant repeated every 10 slots
                v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
                v_K1 := 5;
            }
            case(kHz120) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
                v_K1 := 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    
    //Calculate T3
    v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;  //@sic R5s210842 change 2 sic@
    v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
    
    //T1: send DL Msg
    SRB.send(v_SRB_COMMON_REQ);

    //Enable HARQ ack/nack reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
    
    //T3 configure SS with periodic UL Grant at T3
    f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, p_FreqDomainSchedul, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
                                          
    // Start WatchDog timer
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
    t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 50) / 1000.0);  /* 50ms added to be at the safe side */
    
    alt {
        /* HARQ NACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
            v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
            if (v_DL_RetransmissionCnt > 4) {
                t_Watchdog.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
            } else {
                repeat;
            }
        }
        /* HARQ ACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
            v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
            f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable), -, tsc_NoCnfReq);    /* disable indication of HARQ ack/nack; @sic R5s220683 sic@ */
            v_T2Flag := true;
            repeat;
        }
        /* HARQ error indication */
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        [] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {
            v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
            v_T4Flag := true;
            t_Watchdog.stop;
        }
        [] t_Watchdog.timeout {                                                                                     /* timeout => after T5 */
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
        }
    }
    if(isvalue(nrdc_pdusession)) //WA#WI=988940
    {
        if(nrdc_pdusession)
        {
        SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2,
                                      cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedNR_SRB_COMMON_IND;
        if (not f_Check_NG_PDUSessionModificationComplete (v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].Pdu, v_GSM_MobilityInfo.SessionId)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Modification Complete failed");
        }
        }
    }
    //Stop UL Grant
    f_NR_ULGrantConfiguration_Stop(p_NR_CellId);
    
    if (not v_T2Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
    }
    if (not v_T4Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL Msg not received");
    }
    
    //Check the Delay
    v_T2Slots := (((v_T2.SFN.Number * 10) + v_T2.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T2) + v_MaxNumberSlots;
    v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
    //Initialise Delta2
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_Delta2 := 0;
    } else { //TDD
        select(v_SCS) {
            case(kHz15) {
                v_Delta2 := (5 - (v_T2Slots - v_K1 + v_Nslots)) mod 5;
            }
            case(kHz30) {
                v_Delta2 := (4 - (v_T2Slots - v_K1 + v_Nslots) mod 10) mod 10;
            }
            case(kHz120) {
                v_Delta2 := (5- (v_T2Slots - v_K1 + v_Nslots) mod 5) mod 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    if ( ((v_T4Slots - v_K2) - (v_T2Slots - v_K1))mod v_MaxNumberSlots <= (v_Nslots + v_Delta2)) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
    }

   return v_ReceivedNR_SRB_COMMON_IND;
  }
  



2.2 [bookmark: _Toc126845576][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Function f_TC_8_2_6_2_2_NR5GC_TestBody ()
	Function name
	f_TC_8_2_6_2_2_NR5GC_TestBody ()

	Reason for change
	1) In the current Implementation , at step 2 reconfiguration message is sent to add NR PSCell with SCG DRB.

              While sending reconfiguration message , SKCounter value has to be filled 

2) f_NR5GC_RRC_ProcedureDelay is passed with nrdc_pdusession = true specific to test case 8.2.6.2.2 to handle the Pdu Session Modification Complete ; this is done corresponding to the reasoning given in section 2.1

3) handling of Pdu Session Modification Complete, is moved inside the function f_NR5GC_RRC_ProcedureDelay due to differences in UE behaviour 

4) UL grant is stopped inside the function f_NR5GC_RRC_ProcedureDelay which is not started again to handle the remaining procedures in the test case. so Ul grant has to be started again once out of the function f_NR5GC_RRC_ProcedureDelay().



	Summary of change
	1) At Step 2, while sending reconfiguration message , SKCounter value has to be filled

2) f_NR5GC_RRC_ProcedureDelay is passed with nrdc_pdusession = true

3) handling of Pdu Session Modification Complete is removed from the function f_TC_8_2_6_2_2_NR5GC_TestBody and moved inside the function f_NR5GC_RRC_ProcedureDelay

4)  Ul grant is started again once finished with f_NR5GC_RRC_ProcedureDelay()


	TTCN module
	8_2_6\RRC_Processing_delay_NRDC_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	   function f_TC_8_2_6_2_2_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();
    var template (omit) RadioBearerConfig v_RadioBearerConfig1 := omit;
    var template (omit) RadioBearerConfig v_RadioBearerConfig2 := omit;
    var template (omit) CellGroupConfig v_CellGroupConfigMCG := omit;
    var template (value) CellGroupConfig v_CellGroupConfigSCG;
    var template (omit) NR_RadioBearerList_Type v_SS_ConfigMCG := omit;
    var template (omit) NR_RadioBearerList_Type v_SS_ConfigSCG := omit;
    var template (value) SDAP_Config v_SDAP_ConfigSCG;
    var integer v_SCG_DRB_Id := f_NR_GetNextSCGDRBId();
    var boolean v_IsFR1_SCG := f_NR_CellInfo_GetIsFR1(nr_Cell10);
    var template (value) DRB_ToAddMod v_DRBToAddMod_SCG;
    var template (value) SDAP_Configuration_Type v_SDAP_ConfigurationSCG;
    var RNTI_Value v_RNTI_ValueSecondary :=  tsc_C_RNTI_Value4;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);
    var template (value) QoS_Flow v_QoSFlow;
    var template (value) NG_NAS_DL_Pdu_Type v_NG_NAS_DL_Pdu;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var template (value) NR_RRC_MSG_Request_Type v_RrcMsg;

    // @siclog "Step 1" siclog@
    // SS stops default uplink grant transmission.
    f_NR_ULGrantConfiguration_Stop(nr_Cell1);

    //@siclog "Step 2-4" siclog@
    //The SS transmits an RRCReconfiguration message and a PDU SESSION MODIFICATION COMMAND to add NR PSCell with SCG DRB.
    //The SS transmits one uplink grant after (16*2microsec+1+TDL) slots after step 2 to schedule PUSCH.
    //Check: Does the UE transmit an RRCReconfigurationComplete message within (16*2microsec+1+TDL+K2) slots after successful completion of step 2?
    
    if (isvalue(v_GSM_MobilityInfo)) {
      v_QoSFlow := cs_QoS_Flow5;
      v_QoSFlow.qfi := '000101'B;  // QFI = 5
      v_NG_NAS_DL_Pdu := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId, -, -, -, -, -, f_BuildDefaultQoSRules("5", '7A'O), cs_QoSFlowDescr({v_QoSFlow}));
      // SDAP parameter for the new SCG DRB Id
      // The SDAP needs to include the QFI of the QoS Flow of the SCG bearer.  This will never be a default bearer
      v_SDAP_ConfigSCG := cs_SDAP_Config(oct2int(valueof(v_GSM_MobilityInfo.SessionId)), false, {5});
      v_SDAP_ConfigurationSCG:= f_ReturnNR_SDAP_Configuration_FromSDAP_Config(valueof(v_SDAP_ConfigSCG));
      v_DRBToAddMod_SCG := cs_38508_NRDRB_ToAddMod(v_SDAP_ConfigSCG, v_SCG_DRB_Id, cs_38508_PDCP_Config);
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No internet PDU session found to modify");
    }
    
    v_RadioBearerConfig2 := cs_38508_RadioBearerConfigDef(omit, {v_DRBToAddMod_SCG}, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_SS_ConfigSCG := { cs_NR_SS_DRB_Config(v_SCG_DRB_Id,
                                            cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsDRB),
                                            cs_NR_RlcBearerConfig_AM(v_IsFR1_SCG, -, f_NR_LogicalChannelId_DRB(v_SCG_DRB_Id)),
                                            v_SDAP_ConfigurationSCG) };

    v_CellGroupConfigSCG := f_NR_GetSCGConfigNRDC_SCG(nr_Cell10, v_SCG_DRB_Id, v_RNTI_ValueSecondary);

    // Build info for SCG and store in DRB list
    fl_StoreSCG_InDRBList (valueof(v_DRBToAddMod_SCG), valueof(v_SS_ConfigSCG[0]), valueof(v_CellGroupConfigSCG));

    // Calculate security keys for SCG
    v_SecurityParams := f_NR_Security_InitSCG(v_SecurityParams);
    f_NR_Security_Set(v_SecurityParams);

    f_ReconfigureSS_DRBs_NRDC( nr_Cell1, nr_Cell10, -, v_SS_ConfigMCG, v_SS_ConfigSCG, MCG_SCG_SecondaryKeys );

    v_V1530Ext := f_NRDC_BuildRRCReconfiguration_v1530( v_CellGroupConfigSCG, v_CellGroupConfigMCG, v_RadioBearerConfig2 );
    v_RrcMsg := cs_NR_RRC_MSG_Request_DCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig1, -, omit, v_V1530Ext));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_RrcNasPduList_REQ(nr_Cell1, tsc_NR_RbId_SRB1, -, v_RrcMsg, { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu) }, -, -, cs_RlcBearerRouting_NR(nr_Cell1)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete(cr_RRCReconfigurationComplete_v1530(-, cr_RRCReconfigurationComplete_v1560(?)))),
                               16, "Step 4");

    //@siclog "Step 5" siclog@
    //The UE transmits a ULInformationTransfer message and an PDU SESSION MODIFICATION COMPLETE message.
    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1,
                                      cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
    
    if (not f_Check_NG_PDUSessionModificationComplete (v_ReceivedAsp.Signalling.Nas[0].Pdu, v_GSM_MobilityInfo.SessionId)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Modification Complete failed");
    }
  }



After Change:
	function f_TC_8_2_6_2_2_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();
    var template (omit) RadioBearerConfig v_RadioBearerConfig1 := omit;
    var template (omit) RadioBearerConfig v_RadioBearerConfig2 := omit;
    var template (omit) CellGroupConfig v_CellGroupConfigMCG := omit;
    var template (value) CellGroupConfig v_CellGroupConfigSCG;
    var template (omit) NR_RadioBearerList_Type v_SS_ConfigMCG := omit;
    var template (omit) NR_RadioBearerList_Type v_SS_ConfigSCG := omit;
    var template (value) SDAP_Config v_SDAP_ConfigSCG;
    var integer v_SCG_DRB_Id := f_NR_GetNextSCGDRBId();
    var boolean v_IsFR1_SCG := f_NR_CellInfo_GetIsFR1(nr_Cell10);
    var template (value) DRB_ToAddMod v_DRBToAddMod_SCG;
    var template (value) SDAP_Configuration_Type v_SDAP_ConfigurationSCG;
    var RNTI_Value v_RNTI_ValueSecondary :=  tsc_C_RNTI_Value4;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);
    var template (value) QoS_Flow v_QoSFlow;
    var template (value) NG_NAS_DL_Pdu_Type v_NG_NAS_DL_Pdu;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var template (value) NR_RRC_MSG_Request_Type v_RrcMsg;

    // @siclog "Step 1" siclog@
    // SS stops default uplink grant transmission.
    f_NR_ULGrantConfiguration_Stop(nr_Cell1);

    //@siclog "Step 2-4" siclog@
    //The SS transmits an RRCReconfiguration message and a PDU SESSION MODIFICATION COMMAND to add NR PSCell with SCG DRB.
    //The SS transmits one uplink grant after (16*2microsec+1+TDL) slots after step 2 to schedule PUSCH.
    //Check: Does the UE transmit an RRCReconfigurationComplete message within (16*2microsec+1+TDL+K2) slots after successful completion of step 2?
    
    if (isvalue(v_GSM_MobilityInfo)) {
      v_QoSFlow := cs_QoS_Flow5;
      v_QoSFlow.qfi := '000101'B;  // QFI = 5
      v_NG_NAS_DL_Pdu := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId, -, -, -, -, -, f_BuildDefaultQoSRules("5", '7A'O), cs_QoSFlowDescr({v_QoSFlow}));
      // SDAP parameter for the new SCG DRB Id
      // The SDAP needs to include the QFI of the QoS Flow of the SCG bearer.  This will never be a default bearer
      v_SDAP_ConfigSCG := cs_SDAP_Config(oct2int(valueof(v_GSM_MobilityInfo.SessionId)), false, {5});
      v_SDAP_ConfigurationSCG:= f_ReturnNR_SDAP_Configuration_FromSDAP_Config(valueof(v_SDAP_ConfigSCG));
      v_DRBToAddMod_SCG := cs_38508_NRDRB_ToAddMod(v_SDAP_ConfigSCG, v_SCG_DRB_Id, cs_38508_PDCP_Config);
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No internet PDU session found to modify");
    }
    
    v_RadioBearerConfig2 := cs_38508_RadioBearerConfigDef(omit, {v_DRBToAddMod_SCG}, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_SS_ConfigSCG := { cs_NR_SS_DRB_Config(v_SCG_DRB_Id,
                                            cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsDRB),
                                            cs_NR_RlcBearerConfig_AM(v_IsFR1_SCG, -, f_NR_LogicalChannelId_DRB(v_SCG_DRB_Id)),
                                            v_SDAP_ConfigurationSCG) };

    v_CellGroupConfigSCG := f_NR_GetSCGConfigNRDC_SCG(nr_Cell10, v_SCG_DRB_Id, v_RNTI_ValueSecondary);

    // Build info for SCG and store in DRB list
    fl_StoreSCG_InDRBList (valueof(v_DRBToAddMod_SCG), valueof(v_SS_ConfigSCG[0]), valueof(v_CellGroupConfigSCG));

    // Calculate security keys for SCG
    v_SecurityParams := f_NR_Security_InitSCG(v_SecurityParams);
    f_NR_Security_Set(v_SecurityParams);

    f_ReconfigureSS_DRBs_NRDC( nr_Cell1, nr_Cell10, -, v_SS_ConfigMCG, v_SS_ConfigSCG, MCG_SCG_SecondaryKeys );
      

    v_V1530Ext := f_NRDC_BuildRRCReconfiguration_v1530( v_CellGroupConfigSCG, v_CellGroupConfigMCG, v_RadioBearerConfig2,-,0 ); //WA#WI=988940
    v_RrcMsg := cs_NR_RRC_MSG_Request_DCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig1, -, omit, v_V1530Ext));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_RrcNasPduList_REQ(nr_Cell1, tsc_NR_RbId_SRB1, -, v_RrcMsg, { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu) }, -, -, cs_RlcBearerRouting_NR(nr_Cell1)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete(cr_RRCReconfigurationComplete_v1530(-, cr_RRCReconfigurationComplete_v1560(?)))),
                               16, "Step 4", -,true);

    //@siclog "Step 5" siclog@
    //The UE transmits a ULInformationTransfer message and an PDU SESSION MODIFICATION COMPLETE message.
    /*SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1,
                                      cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
    
    if (not f_Check_NG_PDUSessionModificationComplete (v_ReceivedAsp.Signalling.Nas[0].Pdu, v_GSM_MobilityInfo.SessionId)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Modification Complete failed");
    }*/
    
    f_NR_ULGrantConfiguration_Start(nr_Cell1); //WA#WI=988940

   
      

  }


1. [bookmark: _Toc54888065][bookmark: _Toc126845577][bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]Branches executed
This NR TC was executed in NR-DC FR1+FR2 Band Combination DC_n79A_n257A with NEA1 and NIA1 ciphering and integrity algorithms.

4 [bookmark: _Toc126845578]Execution Log Files 
4.1 [bookmark: _Toc126845579]Snapdragon X70 5G Modem-RF System
The Snapdragon X70 5G Modem-RF System, Qualcomm Technologies’ 5th generation modem-to-antenna 5G solution passed this test case on R&S® 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
Test case execution log file:
TC_8_2_6_2_2_Qualcomm.log
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file the complete test case execution as well as the PICS/PIXIT parameter settings can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
5 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc126845580]References
	[1]
	R5s230159   Supporting information for addition of NR-DC test case 8.2.6.2.2 in FR1+FR2




